
Getting Started with Maple
From materials by Warren Weckesser, Colgate University

This file is a "Maple Session."  The actual Maple commands that were entered are 
shown in the lines preceded by the ">"; To execute a command, place your cursor 
after the semicolon (or anywhere else in the "execution block") and press 
"Return/Enter". 

Feel free to experiment with your own calculations, using the blank command lines 
that follow each example.  You can add additional command lines by pressing Ctrl+J 
to get a ">" prompt after the current cursor position and Ctrl+K to get one before 
(for Macs it's Command+J and Command+K). 

Simple calculations: (click the triangle to open the section)
2+2;

Note the semicolon at the end of the line.  Each Maple command must end with a semicolon (or a 
colon if you don't want the output to be displayed).

(12+3*5)/4;

Note that Maple does not convert a rational number to its decimal representation.  You can find the 
decimal representation with the "evalf" command.

evalf(27/4);

Maple has many predefined functions.  For example, sine and cosine:
sin(1.23);

cos(0);

cos(1);
Note that Maple didn't display the numerical value of cos(1).  To see the numerical value, use 
"evalf":

evalf(cos(1));

To take a square root, you can use the "sqrt" function:
sqrt(4);

Maple has several predefined constants.  The one that we will need most is pi.  The Maple name of 
pi that can be used in an input line is "Pi".  Maple is case sensitive, so the P must be a capital letter.  
In the output, Maple will show the Greek letter.

Pi;
evalf(Pi);

sin(Pi);

Another predefined constant that you will see occasionally is "I".  In Maple, "I" is the square root of
-1 (a complex number).   For example: 

sqrt(-1);
sqrt(-16);
I*I;
(3+4*I)/(5+6*I);

The function name "exp" is used for the exponential function:
exp(0);
exp(1);

Note that the number "e" is displayed in the output as a boldface e.  However, do not use e^x for 
the exponential function; use exp(x).

evalf(exp(1));
By default, "evalf" gives us ten decimal places. We can adjust this:

evalf(exp(1),2);
or alternatively,

evalf[2](exp(1));
evalf[1000](Pi);

Making assignments
Typically we want to do a lot more than calculate a few numbers.  We want to define functions, plot
graphs, solve equations, etc.  We need to be able to define and store expressions in Maple.  This is 
done with the assignment statement ":=".  For example:

m := 355/113;
Now the variable "m" has been created, and it has the value 355/113.  We can now use "m" in 
subsequent commands.

evalf(m);
sin(m);

The value of a variable can be an expression that is more complicated than just a number. For 
example, suppose we want to work with the expression 4x^2-1.  We can create a variable that will 
hold this expression.  I will call the variable "q":

q := 4*x^2-1;

WARNING: In this worksheet: don't reassign new expressions to existing variables!. This may 
cause subsequent calculations using these variables to fail!

Now there are many things we can do with the expression q.  For example, we can plot it, using the 
"plot" command.

plot(q,x=-1..1);

We'll take a closer look at the plot command later. 

We can also take the derivative with respect to x, using the "diff" command.  I will create a new 
variable called "dq" that holds the derivative:

dq := diff(q,x);
We could have done the same things without defining q.  We can give the expression itself as an 
argument to the functions:

plot(4*x^2-1,x=-1..1);
diff(4*x^2-1,x);

We can also attempt to solve the equation 4x^2-1=0 with the "solve" command:
solve(4*x^2-1=0,x);

In this case, the solve command was able to find the solutions. 

Exercise: In the workspace below, use "plot", "diff", and "solve" to carry out the same steps on the 

quadratic x2C xK2. 

Closer look at "plot"
The "plot" command takes several arguments, and has many options.  The basic command takes two
arguments. The first is the expression (i.e. function) to be plotted, and the second is the range of the 
independent variable of the expression.  For example, the following command plots x^2 on the 
interval -1 < x < 1:

plot(x^2,x=-1..1);
Note that a..b is the standard method for describing a range in Maple.

We can also give a third argument to the plot command that specifies the vertical range of the plot.  
For example:

plot(x^2,x=-1..1,-1..1.5);
You can specify more optional arguments.  In the next example, I've changed the color to black, 
used a dashed line, and I've given the plot a title:

plot(x^2,x=-1..1,-1..1.5,title=`y=x^2`,color=black,linestyle=
DASH);

To see more examples, look up "plot" in the Maple Help,  or enter the command "?plot".

Closer look at "diff"
You can take derivatives in Maple with the "diff" command ("diff" stands for "differentiate").  The 
basic command takes two arguments. The first is the expression to be differentiated, and the second 
is the variable with respect to which the derivative is to be taken.  Maple knows the derivatives of 
most of the elementary functions.

Here are some examples.  The first finds the derivative of sin(t)+4t, and the second finds the 
derivative of 1/x + exp(x^2).

diff(sin(t)+4*t,t);
diff(1/x + exp(x^2),x);

To take a second or higher order derivative, just list the variable again.  Here are some second 
derivatives:

diff(sin(t)+4*t,t,t);
diff(1/x + exp(x^2),x,x);

 "solve" (closer look) & "complexplot"
The "solve" command can solve many algebraic equations.  We saw an example earlier where we 
solved a quadratic.  In the simplest form of the command, the first argument is the equation to be 
solved, and the second is the variable to be solved for.   Here we find the solutions to x^2+5x+6=0:

solve(x^2+5*x+6=0,x);
If there expression does not contain an equals sign, Maple assumes that the expression should be set
equal to zero.

solve(x^2+5*x+6,x);
Next we solve a cubic polynomial:

solve(x^3-2*x+1=0,x);
Let's make one small change to the cubic polynomial:

solve(x^3-x+1=0,x);
What a mess!  If you look carefully at the above output, you will see that Maple has found three 
solutions, two of which are complex numbers.  (Recall that the symbol I in Maple is the complex 
number sqrt(-1).) 

Here "evalf" comes to our rescue. 
evalf(%);

Here the symbol % is used to recall the output of the last calculation (likewise, %%  recalls the 
output of the next-to-last calculation ...). However it's not a good habit to rely on %   ... much better 
to get in the habit of assigning each output a name, like this:

solns:=solve(x^3-x+1=0,x);
fsolns:=evalf(solns): #No displayed output because of colons 
                      #Use "#" for inline comments to help you 
                      #document your work.
fsolns;

We can use "complexplot" to plot complex solutions:
with(plots);  #loads the advanced plotting package "plots".
              #follow by colon instead of semicolon to suppress
long list
complexplot([fsolns], style=point, symbol=circle);

Question: What happens if you leave out the "style=point" option?

For more information about "complexplot", just use Maple's help command:
?complexplot

and pay particular attention to the examples at the end of the page. The same goes for any Maple 
command! Try this on "solve" and "evalf" for example.

What happens if the equation has no solution, or if Maple can not find it?  Here are some examples. 
First, a trivial example with no solution:

solve(1=0,x);
Maple gave no output; this means that either there is no solution, or Maple can not find any 

Note that Maple can solve an equation like x^2=-1, because Maple allows complex numbers:
solve(x^2=-1,x);

Sometimes Maple gives a partial solution:
partialsoln:=solve(x*(sin(x)+x+1/4)=0,x);

The "RootOf" function means the root of the expression; in the expression, _Z is the variable.  In 
this example, Maple found one solution to be 0, but the other "solution" is just the solution to 4*_Z 
+ 4*sin(*_Z) +1=0.  Maple could not find an analytical solution to this equation. However, it can 
provide a decimal approximation:

evalf(partialsoln);
To see what's really going on with this function, try plotting it using the "plot" command, selecting 
various intervals about the origin in order to see if there are other solutions.

A max-min problem via Maple
Let's create a function, plot it, and then use "diff" and "solve" to find the critical points.

w := (x^4+3*x^2)/(1+x^4);
plot(w,x=-3..3);
dw := diff(w,x);
plot(dw,x=-3..3);
critpts:=solve(dw=0,x);

Note that two of the five solutions found by "solve" are actually complex numbers.

Exercise: (a) display these solutions as decimals using "evalf".
                (b) plot these complex numbers using "complexplot."


